
Making Customized ICH Listings with ODS RTF
Huei-Ling Chen, William Wei, Merck & Co., Inc.

ABSTRACT
ICH (International Consortium on Harmonization) data listings are common reports prepared by
pharmaceutical companies in regulatory submission. Such listings are often submitted in RTF format.
This paper presents a technique that can be used to efficiently produce a customized ICH Abnormal Lab
Listing based on a company’s uniform RTF output standards. This technique includes five components:
data preparation; and four SAS® syntaxes (SYSTEM OPTIONS, ODS TAGSETS.RTF, PROC
TEMPLATE style, and PROC REPORT) can be utilized to define the layout and to render a data listing
table. This paper will first describe the problem that we are trying to solve and then give details on the
ODS TAGSETS.RTF options that we have chosen to solve this problem with mock data.

INTRODUCTION
Data listing at first glance is just a simple listing that can be fulfilled using PROC PRINT procedure or
simple PROC REPORT procedure. However, there are certain data formats and display guidelines that
need to be followed, when it is used for submission to a regulatory agency, for example, ICH listing. The
International Consortium on Harmonization provides a guiding document, ICH Guideline E3, specifically
on the content of a CSR.
Below is a snapshot of the ICH E3 guideline on abnormal lab data listing.

According to the ICH E3 guideline, pharmaceutical companies design the outlook for each ICH data
listing, and make sure that the listing fulfills the requirements for data formats and display principles. For
example, data are grouped by investigator or not. Decide how and where to present the treatment arm
and subject information, on top of the table or present it as individual columns inside the table body. For
abnormal lab finding listing, the data should be displayed by laboratory category alphabetically or be
based on the order of importance of the tests.

In order to create a listing based on the ICH E3 guideline while also satisfying a company’s uniform RTF
output standards, We have developed a technique that integrates the following five components to
produce the desired output: thoughtful data preparation based on the desired table contents, and use
SAS options/procedures (SYSTEM OPTIONS, ODS TAGSETS.RTF, PROC TEMPLATE style, and
PROC REPORT) to tweak table layout and appearance. Novel conglomeration of these components
creates Table 1 as the RTF output.

Output 1: ICH Abnormal Lab Listing using mock up data

1. DATA PREPARATION WORK
The required data is assembled by integrating information from three different domains: LB provides the
actual abnormal lab readings and its associated lab information; DM provides details on the subject who
originated the abnormal lab readings; VS provides vital sign information such as weight of the subject
who originated the abnormal lab readings.
Display 1 is the mock up data for Output 1 in original data form.

Display 1: mock up data for Output 1 (original data)

Group the data by the investigator and place the investigator site information as a ‘site’ title row at each
page. Two headlines are inserted and present the treatment arm and subject information. To carry out
these tasks, three variables are added to the dataset. See the orange box columns in Display 2.
• A site title row (variable name: Study_Site_Line)
o Variable value example: “Trial Number: 99999-999, Site: 0001, Investigator: Doe, John,
Country: United States”
• A line row with treatment arm text name (variable name: Decode_1)
o Variable value example: “ISG”
• A line row with demographic data for each subject row (variable name: Subj_Detail _Line)
o Variable value example: “Trial Number=99999-999, Site Number=0001, Unique Subject
ID=99999-999_000900001, Gender=M, Age=51 Years, Weight=120 LB”
Instead of displaying the listing alphabetically, sometimes it is necessary to add numeric display ordering
variables in the COLUMN statement using PROCE REPORT to indicate the right listing order. In this
paper, five variables are added. See the green box columns in Display 2.
• A treatment arm order variable with 1 to 1 match to variable Decode_1 (variable name: Coded_1)
o Missing out this numeric variable, the data listing will be sorted alphabetically by the
treatment arm text name.
• An order variable with 1 to 1 match to variable Subj_Detail_Line (variable name: Sec_No1)
• An order variable with 1 to 1 match to variable LBCAT (variable name: Sec_No2)
• An order variable with 1 to 1 match to variable LBTEST (variable name: Sec_No3)
• An order variable to sort the data by collection date (variable name: Sortdt)
o Missing out this numeric variable, the data listing will be sorted alphabetically by the
Epoch name, instead of the right collection date.
Two dummy variables are added for placing the footnotes at bottom of the table. See the blue box
columns in Display 2.
• Dummy variable 1 (variable name: FullSetIn)
• Dummy variable 2 (variable name: FullSetOut)
Display 2 is the complete data feeding into Proc Report procedure to create Output 1.

Display 2: mock up data for Output 1 (complete data for Proc Report procedure)

2. SYSTEM OPTIONS TO SETUP THE PAGE LAYOUT AND THE WORKING
ENVIRONMENT
OPTIONS cbufsize=65535 ORIENTATION=landscape nodate nobyline number
topmargin=1.25in bottommargin = 1in leftmargin=0.5in
rightmargin=0.5in;
•
•
•

nodate: suppress date information in the display
nobyline: suppress the automatic printing of BY lines above each BY group
Cbufsize: increase the buffer size to handle large data

When dealing with huge data, the following error message may appear.
“ERROR: The SAS System stopped processing this step because of insufficient
memory.”
This issue can easily be resolved.by increasing the buffer size using “cbufsize=65535” in OPTIONS
statement

3. PROC TEMPLATE STYLE TO DEFINE APPEARANCE FOR TABLE COMPONENTS AND
CONTENTS
When using PROC TEMPLATE procedure to define the style, there is no need to define every single
element from scratch. Rather, we can take an existing style template, such as STYLES.RTF, as the
parent style and only modify a few styles according to the preferred table format, for example, fonts,
colors, table margins, page number, etc.
proc template;
define style mytemplate;
parent = styles.rtf;
< style element syntax>
run;

Style Element Syntax

replace fonts /
'TitleFont2' = ("Times New Roman",8pt)
'TitleFont' = ("Times New Roman",12pt)
'StrongFont' = ("Times New Roman",8pt)
'EmphasisFont' = ("Times New Roman",8pt)
'FixedEmphasisFont' = ("Times New Roman,Courier",8pt)
'FixedStrongFont' = ("Times New Roman,Courier",8pt)
'FixedHeadingFont' = ("Times New Roman,Courier",8pt)
'BatchFixedFont' = ("Times New Roman, Courier",8pt)
'FixedFont' = ("Times New Roman,Courier",8pt)
'headingEmphasisFont' = ("Times New Roman",8pt)
'headingFont' = ("Times New Roman",8pt)
'docFont' = ("Times New Roman",8pt);
replace color_list "Colors used in the default style" /
'bg' = _undef_
'fg' = black
'bgH' = _undef_
'link' = blue;
replace table from Output /
borderwidth = .5pt
borderspacing = 0pt
cellpadding = 0.5pt
rules = group
frame = box;
replace Body from Document "Controls the Body file." /
marginbottom = 1.25in
margintop = 2in
marginright = 0.5in
marginleft = 0.5in;
replace pageno from pageno /
fontsize = 12 pt
cellpadding = 0
cellspacing = 0
pretext="Page "
posttext=" of (*ESC*){lastpage} ";
class SystemFooter /
font = ("Times New Roman",8pt);

Notes
The fonts of the tables are
defined within the style
element Fonts. In the
table, fonts ‘Time New
Roman’ is specified.

Undefine table
background and table
header background color.
This setting exposes the
water marker.
RULES=GROUP will
place a divider between
the Header and the table
area of the table.
FRAME=BOX option will
draw only the outside
border lines.
Set up the table margins.

Display page number in
header area of rtf table.

Customize the footer by
adding the SystemFooter
style element.

Note that when using some table row margin control or parskip in the template, it is necessary to pay
attention to the page break in large output with tagset.rtf. Some combinations of settings may cause the
page break abnormally. If this case happens, the cell padding/parskip option needs to be adjusted to
adapt the page break.

4. ODS TAGSETS.RTF TO SPECIFY USING RTF SYNTAX WITH PREDEFINED TEMPLATE
TO GENERATE LISTING TABLE
This paper utilizes settings in the SAS® Output Deliver System (ODS) destinations RTF tagset.
ods noresults escapechar='~';
ods listing close;
ODS tagsets.rtf
FILE="C:\ProcReport\output\ichlablisting.rtf"
style = mytemplate
uniform
tablerows=25
options(continue_tag="no" watermark=”xxx”)

bodytitle;
< PROC REPORT syntax >
ODS tagsets.rtf CLOSE;
ODS listing;
The ODS TAGSETS.RTF provides the formats that are most frequently used by the pharmaceutical
industry. ODS TAGSETS.RTF have several advantages over traditional ODS RTF in that it offers more
controls over Table of Contents, and titles/footnotes could be repeated on each page as a body type
instead of going to the header/footer area in a Word document.
• style: Specify the template to use in the RTF output
• uniform: Specify that every page of a table is formatted the same
• tablerows: Specify the maximum number of table rows in a page
• options(continue_tag=”no”): Do not add continue tag when a table break and continue to the
next page
• options(watermark=”xxx”): Add watermark at the table background, such as “CONFIDENTIAL”.
• bodytitle: Specify the table titles and footnotes are displayed in the RTF body instead of in the
header and footnote area.

5. PROC REPORT TO RENDER TABLE CONTAINT WITH MORE FORMATING
PROC REPORT DATA=temp SPACING=1 LS=118 PS=20 SPLIT='|' MISSING NOWD
spanrows
style(report)= . . .
STYLE(HEADER)= . . .
BY study_site_line;
COLUMN FullSetOut FullSetIn study_site_line coded_1 decode_1 sec_no1
subj_detail_line . . .
DEFINE
DEFINE
DEFINE
DEFINE
DEFINE
DEFINE
DEFINE
. . .

FullSetOut /order noprint "" LEFT ;
FullSetIn /order noprint "" LEFT ;
study_site_line / ORDER ORDER=INTERNAL noprint "" ;
coded_1 / ORDER ORDER=INTERNAL noprint "" ;
decode_1 / ORDER ORDER=INTERNAL noprint "" ;
sec_no1 / ORDER ORDER=INTERNAL noprint "" ;
subj_detail_line / ORDER ORDER=INTERNAL noprint "" ;

< COMPUTE BLOCKS >
RUN;

CREATING A TABLE WITH OUTSIDE BORDER LINE, INSIDE VERTICAL BORDER
LINE, NO INSIDE HORIZONTAL BORDER LINE
Proc Template style uses frame = box option to draw only the outside border lines. The inside vertical
border line can be defined in Proc Report syntax using style (column) option in the define statement of
Proc Report syntax.
Proc Template
Style

proc template;
define style mytemplate;
parent = styles.rtf;
replace table from Output /
borderwidth = 0.5pt
borderspacing = 1pt
cellpadding = 2pt

rules = group
frame = box;

Proc Report

PROC REPORT . . .
COLUMN . . .
DEFINE LBORNRLO/ ORDER ORDER=INTERNAL "" LEFT
style(column)={borderleftcolor=black borderleftwidth=1
borderrightcolor=black borderrightwidth=1
ASIS=ON };
DEFINE LBORNRHI/ ORDER ORDER=INTERNAL "" LEFT
style(column)={borderleftcolor=black borderleftwidth=1
borderrightcolor=black borderrightwidth=1
ASIS=ON };
Run;

Output 2: A table with inside vertical border line but no inside horizontal border line

ERASING THE BORDER LINE BETWEEN HEADERS
Notice the column normal range (low, high) has two lines of header. The text ‘Normal Range‘ is on the
top header line. The texts, ‘Low’ and ‘High’, are on the bottom header line. There are no inside border
line between these headers. The in-line formatting code is used here to erase the inside border lines
between these three head texts. Choose the escape character tilde (~) symbol. Another commonly
used character is hat (^) symbol. Since some laboratory test unit ^ symbol appears in the data, a safer
choice, tilde symbol, is used as escape character. The borderbottomcolor=white and
borderbottomwidth=.1pt make the line white color and minimize the line width.
Proc Report

PROC REPORT . . .
COLUMN . . .
("~S={borderbottomcolor=white borderbottomwidth=.1pt}Normal Range"
("~S={bordertopcolor=white bordertopwidth=.1pt
borderrightcolor=white borderrightwidth=.1pt}Low" LBORNRLO)
("~S={bordertopcolor=white bordertopwidth=.1pt
borderleftcolor=white borderleftwidth=.1pt}High" LBORNRHI) )
DEFINE LBORNRLO/ . . .
DEFINE LBORNRHI/ . . .
Run;

DISPLAYING A SITE TITLE ROW BEFORE THE HEADER ROW
By ICH E3 requirement, data should be sorted and grouped by investigators. Based on this requirement
and company’s standard, a site title row is used to display the investigator site information at the top of
the body section on each page right before the header row.
A site title row variable study_site_line is added to the dataset, as mentioned before. Use COMPUTE
BLOCK feature in the Proc Report procedure. We can specify to display that variable at the LOCATION
of BEFORE the REPORT_ITEM of _PAGE_ to print out the investigator site information at the top of each
page.
Data Preparation
Work

Add site title row variable study_site_line to the dataset.

Proc Report

PROC REPORT . . .
BY study_site_line;
COMPUTE BEFORE _PAGE_ / left style={font_size=12pt };
line "~R'\b '" study_site_line $120.;
line "";
ENDCOMP;
Run;

Output 3: Displaying a site title row before the header row

INSERTING LINE ROWS WITH TREATMENT ARM AND DEMOGRAPHIC DATA FOR
EACH SUBJECT
Based on ICH requirement and company’s standard we also need to have two headlines containing
subject demographic data before of the records of each patient. The required demographic data for each
subject includes therapy group, protocol number, site number, subject identifier, gender, age, and weight.
Two variables containing the headlines information are added to the dataset.
Again, we can create two COMPUTE BLOCKs to display the headline variables at the LOCATION of
BEFORE the corresponding REPORT_ITEMs.
Notice style option is utilized here to draw the horizontal border line for these two headlines.
Data Preparation
Work

Two headline variables, decode_1 and subj_detail_line, are created to
present the treatment arm and subject information variable to the dataset.

Proc Report

PROC REPORT . . .
COMPUTE BEFORE decode_1 /
STYLE={BORDERLEFTWIDTH=1
BORDERTOPWIDTH=1
BORDERRIGHTWIDTH=1
BORDERBOTTOMWIDTH=1};
line "~R' \ql '" decode_1 $200.;
ENDCOMP;

COMPUTE BEFORE subj_detail_line /
STYLE={BORDERLEFTWIDTH=1
BORDERTOPWIDTH=1
BORDERRIGHTWIDTH=1
BORDERBOTTOMWIDTH=1};
line "~R' \ql '" subj_detail_line $200.;
ENDCOMP;
Run;

Output 4: Inserting line rows with treatment arm and demographic data for each subject

DISPLAYING PAGE NUMBER IN THE HEADER AREA
Displaying page numbers can be done in multiple methods. SAS option PAGENO is one method. The
easiest way is to add RTF code with ODS Escape character, Page ~{pageof}, in the title statement.
However, SAS option BODYTITLE places the table title on page content, not in the header area. This
method makes the page number in the page content as well.
In case the page number need to be displayed in the header area, here we demonstrate another method.
Proc Template
Style

proc template;
define style mytemplate;
parent = styles.rtf;
replace pageno from pageno /
fontsize = 12 pt
cellpadding = 0 cellspacing = 0
pretext="Page "
posttext=" of (*ESC*){lastpage} ";

System Options

OPTIONS NUMBER;

The page number in Output 5 is in grey out header area, not on page content.

Output 5: Displaying page number in the header area

TABLE MARGINS
There are multiple ways to set up the margins of the table. Proc Template style uses margintop,
marginbottom, marginleft, and marginright to set up the margins. System options also can do it.
Proc Template
Style

System Options

proc template;
define style mytemplate;
parent = styles.rtf;
replace Body from Document "Controls the Body file." /
margintop = 1.25in
marginbottom = 1in
marginleft = 0.5in
marginright = 0.5in
;
OPTIONS topmargin = 1.25in
bottommargin = 1in
leftmargin = 0.5in
rightmargin = 0.5in ;

To check the table margins, click ‘Layout’  ‘Margins’  ‘Custom Margins’.

Output 6: Checking table margins

TWO DIFFERENT FOOTNOTES, ONE INSIDE THE TABLE BOX, ONE OUTSIDE
TABLE BOX
There is certain standard table design requires two groups of footnotes at the end of the entire listing.
One group of footnotes are for study specific information to be displayed inside of the table frame. The
other group of footnotes are for data source information, to be displayed outside of the table frame.
Two dummy variables with value of 1, can be added to the dataset to facilitate these kinds of footnote
creation.
Once again, we can use COMPUTE BLOCKs to display the footnotes at the LOCATION of AFTER the
REPORT_ITEMs of the value of two dummy variables. Since the records of entire dataset have the same
value of 1, the footnote will be only displayed after all the records, which is the end of the entire table.

The style options are utilized here to draw the lines at necessary place.
There are two benefits of using the compute block to create footnotes, instead of the commonly used
footnote statement. It is much easier to have the footnotes align with table using this method. It also
brings the flexibility of style options to draw border lines.
Data Preparation
Work

Add FULLSETIN and FULLSETOUT variables to the dataset.

Proc Report

PROC REPORT . . .
COLUMN FullSetOut FullSetIn study_site_line coded_1 . . .
COMPUTE AFTER FullSetIn /
STYLE={BORDERLEFTWIDTH=1 BORDERTOPWIDTH=1
BORDERRIGHTWIDTH=1 BORDERBOTTOMWIDTH=1};
LINE "~R'\ql' ~{super ~{unicode '2020'x}}Relative Day ...";
LINE "~R'\ql' ~{super ~{unicode '2021'x}}Indicates this …";
LINE "~R'\ql' Baseline Regimen = …";
...
ENDCOMP;
COMPUTE AFTER FullSetOut/
STYLE={BORDERLEFTWIDTH=0 BORDERTOPWIDTH=0
BORDERRIGHTWIDTH=0 BORDERBOTTOMWIDTH=0};
LINE "Source: ~R'\ql' [P999MK9999A: adam-adsl]
[P999MK9999A: sdtm-lb; supplb; vs; suppvs]";
ENDCOMP;
Run;

Output 7: Footnotes inside/outside table box

DISPLAYING WATERMARK
In some case the watermark could convey message in a specific way. Here in the template, we undefined the table
and header back ground color to ensure the watermark could appear in the back of tables.

Proc Template
Style

proc template;
define style mytemplate;
parent = styles.rtf;
replace color_list "Colors used in the default style" /
'bg' = _undef_
'fg' = black
'bgH' = _undef_

;

ODS
TAGSETS.RTF

'link' = blue;

ODS tagsets.rtf
FILE="C:\ProcReport\output\ichlablisting.rtf"
style = mytemplate
uniform
tablerows=25
options(continue_tag="no" watermark=”xxx”)
bodytitle;
< PROC REPORT syntax >
ODS tagsets.rtf CLOSE;

OTHER TECHNIQUES USED IN THIS MACRO DESIGN:
AUTO-ADJUSTING THE COLUMN WIDTH
If we plan to display all the columns in one page, it is better to automatically assign the column widths according to
user input. Here we used percent unit to specify column width. Depending on the number of columns, practically use
95-99 percent of the page width (not include the margins) is the best fit. For example, each column width is
current_specified_col_width/sum_of_columns*95 pct. In this case, the lines between the columns are considered.

SPECIFYING SPECIAL CHARACTER IN THE TABLE OR FOOTNOTE
In order to display special character, especially to be compatible with long lasting standard system, a new set of
special character sets need to be defined for proc report. In this version we applied escape charrcter directly in the
macro variable assign statement. So, user could call the macro variable directly. For example, the double daggers
superscript (††) is defined as a macro variable as following:

%LET dagger2_ = ~{super ~{unicode '2020'x '2020'x}};
If we pass this special character to proc report procedure, we need to be cautious what mask function is to be used.
Especially when we transfer multiple special characters, such as comma (,), which is a parameter divider, or another
unpaired quotation mark, a proper mask function is needed to be called to make sure the full string was masked and
transferred to proc report correctly.

CONCLUSION
Although these techniques may appear intimidating and tedious, the benefits of efficiently producing a customized
ICH Abnormal Lab Listing with a company’s uniform RTF output standards certainly outweigh the effort. A great
number of companies in the pharmaceutical industry use PROC REPORT for reporting, highlighting the tool’s
reliability and current popularity.

